
Convex.jl is a Julia package for Disciplined Convex Programming. Convex.jl makes it easy to describe optimization
problems in a natural, mathematical syntax, and to solve those problems using a variety of different (commercial and
open-source) solvers. Convex.jl can solve

• linear programs

• mixed-integer linear programs and mixed-integer second-order cone programs

• dcp-compliant convex programs including

– second-order cone programs (SOCP)

– exponential cone programs

– semidefinite programs (SDP)

Convex.jl supports many solvers, including Mosek, Gurobi, ECOS, SCS and GLPK, through the MathProgBase inter-
face.

Note that Convex.jl was previously called CVX.jl. This package is under active development; we welcome bug reports
and feature requests. For usage questions, please contact us via the JuliaOpt mailing list.

CHAPTER 1

In Depth Documentation:

1.1 Installation

Installing Convex.jl is a one step process. Open up Julia and type

Pkg.update()
Pkg.add("Convex")

This does not install any solvers. If you don’t have a solver installed already, you will want to install a solver such as
SCS by running

Pkg.add("SCS")

To solve certain problems such as mixed integer programming problems you will need to install another solver as well,
such as GLPK. If you wish to use other solvers, please read the section on solvers.

1.2 Quick Tutorial

Consider a constrained least squares problem

minimize ‖𝐴𝑥− 𝑏‖22
subject to 𝑥 ≥ 0

with variable 𝑥 ∈ R𝑛, and problem data 𝐴 ∈ R𝑚×𝑛, 𝑏 ∈ R𝑚.

This problem can be solved in Convex.jl as follows:

# Make the Convex.jl module available
using Convex

# Generate random problem data
m = 4; n = 5
A = randn(m, n); b = randn(m, 1)

# Create a (column vector) variable of size n x 1.
x = Variable(n)

# The problem is to minimize ||Ax - b||^2 subject to x >= 0
# This can be done by: minimize(objective, constraints)
problem = minimize(sum_squares(A * x + b), [x >= 0])

# Solve the problem by calling solve!
solve!(problem)

# Check the status of the problem
problem.status # :Optimal, :Infeasible, :Unbounded etc.

# Get the optimum value
problem.optval

1.3 Basic Types

The basic building block of Convex.jl is called an expression, which can represent a variable, a constant, or a function
of another expression. We discuss each kind of expression in turn.

1.3.1 Variables

The simplest kind of expression in Convex.jl is a variable. Variables in Convex.jl are declared using the Variable
keyword, along with the dimensions of the variable.

# Scalar variable
x = Variable()

# Column-vector variable
x = Variable(5)

# Matrix variable
x = Variable(4, 6)

Variables may also be declared as having special properties, such as being

• (entrywise) positive: x = Variable(4, Positive())

• (entrywise) negative: x = Variable(4, Negative())

• integral: x = Variable(4, :Int)

• binary: x = Variable(4, :Bin)

• (for a matrix) being symmetric, with nonnegative eigenvalues (ie, positive semidefinite): z =
Semidefinite(4)

1.3.2 Constants

Numbers, vectors, and matrices present in the Julia environment are wrapped automatically into a Constant expression
when used in a Convex.jl expression.

1.3.3 Expressions

Expressions in Convex.jl are formed by applying any atom (mathematical function defined in Convex.jl) to variables,
constants, and other expressions. For a list of these functions, see operations. Atoms are applied to expressions
using operator overloading. Hence, 2+2 calls Julia’s built-in addition operator, while 2+x calls the Convex.jl addition
method and returns a Convex.jl expression. Many of the useful language features in Julia, such as arithmetic, array

indexing, and matrix transpose are overloaded in Convex.jl so they may be used with variables and expressions just as
they are used with native Julia types.

Expressions that are created must be DCP-compliant. More information on DCP can be found here.

x = Variable(5)
# The following are all expressions
y = sum(x)
z = 4 * x + y
z_1 = z[1]

Convex.jl allows the values of the expressions to be evaluated directly.

x = Variable()
y = Variable()
z = Variable()
expr = x + y + z
problem = minimize(expr, x >= 1, y >= x, 4 * z >= y)
solve!(problem)

# Once the problem is solved, we can call evaluate() on expr:
evaluate(expr)

1.3.4 Constraints

Constraints in Convex.jl are declared using the standard comparison operators <=, >=, and ==. They specify rela-
tions that must hold between two expressions. Convex.jl does not distinguish between strict and non-strict inequality
constraints.

x = Variable(5, 5)
# Equality constraint
constraint = x == 0
# Inequality constraint
constraint = x >= 1

Matrices can also be constrained to be positive semidefinite.

x = Variable(3, 3)
y = Variable(3, 1)
z = Variable()
# constrain [x y; y’ z] to be positive semidefinite
constraint = isposdef([x y; y’ z])

1.3.5 Objective

The objective of the problem is a scalar expression to be maximized or minimized by using maximize or minimize
respectively. Feasibility problems can be expressed by either giving a constant as the objective, or using problem =
satisfy(constraints).

1.3.6 Problem

A problem in Convex.jl consists of a sense (minimize, maximize, or satisfy), an objective (an expression to which the
sense verb is to be applied), and zero or more constraints which must be satisfied at the solution. Problems may be
constructed as

1.3. Basic Types 5

problem = minimize(objective, constraints)
# or
problem = maximize(objective, constraints)
# or
problem = satisfy(constraints)

Constraints can be added at any time before the problem is solved.

# No constraints given
problem = minimize(objective)
# Add some constraint
problem.constraints += constraint
# Add many more constraints
problem.constraints += [constraint1, constraint2, ...]

A problem can be solved by calling solve!:

solve!(problem)

After the problem is solved, problem.status records the status returned by the optimization solver, and can
be :Optimal, :Infeasible, :Unbounded, :Indeterminate or :Error. If the status is :Optimal,
problem.optval will record the optimum value of the problem. The optimal value for each variable x partici-
pating in the problem can be found in x.value. The optimal value of an expression can be found by calling the
evaluate() function on the expression as follows: evaluate(expr).

1.4 Operations

Convex.jl currently supports the following functions. These functions may be composed according to the DCP compo-
sition rules to form new convex, concave, or affine expressions. Convex.jl transforms each problem into an equivalent
cone program in order to pass the problem to a specialized solver. Depending on the types of functions used in the
problem, the conic constraints may include linear, second-order, exponential, or semidefinite constraints, as well as
any binary or integer constraints placed on the variables. Below, we list each function available in Convex.jl orga-
nized by the (most complex) type of cone used to represent that function, and indicate which solvers may be used to
solve problems with those cones. Problems mixing many different conic constraints can be solved by any solver that
supports every kind of cone present in the problem.

1.4.1 Linear Program Representable Functions

An optimization problem using only these functions can be solved by any LP solver.

operation description vex-
ity

slope implicit
constraint /
notes

x+y or x.+y addition affine increasing none
x-y or x.-y subtraction affine increasing in 𝑥

decreasing in 𝑦
none
none

x*y multiplication affine increasing one term is
constant

x/y division affine increasing 𝑦 is a scalar
constant

x .* y elemwise
multiplication

affine increasing one term is
constant

x[1:4, 2:3] indexing and
slicing

affine increasing none

diag(x, k) 𝑘-th diagonal of
a matrix

affine increasing none

x’ transpose affine increasing none
x’*y or dot(x,y) 𝑥′𝑦 affine increasing one term is

constant
vec(x) vector

representation
affine increasing none

reshape(x, m, n) reshape into
𝑚× 𝑛

affine increasing none

minimum(x) min(𝑥) con-
cave

increasing none

maximum(x) max(𝑥) con-
vex

increasing none

[x y] or [x; y]
hcat(x, y) or
vcat(x, y)

stacking affine increasing none

trace(x) tr (𝑋) affine increasing none
min(x,y) min(𝑥, 𝑦) con-

cave
increasing none

max(x,y) max(𝑥, 𝑦) con-
vex

increasing none

pos(x,y) max(𝑥, 0) con-
vex

increasing none

neg(x,y) max(−𝑥, 0) con-
vex

decreasing none

inv_pos(x) 1/max(𝑥, 0) con-
vex

decreasing 𝑥 > 0

abs(x) |𝑥| con-
vex

increasing on 𝑥 ≥ 0
decreasing on 𝑥 ≤ 0

none

1.4.2 Second-Order Cone Representable Functions

An optimization problem using these functions can be solved by any SOCP solver (including ECOS, SCS, Mosek,
Gurobi, and CPLEX). Of course, if an optimization problem has both LP and SOCP representable functions, then any
solver that can solve both LPs and SOCPs can solve the problem.

1.4. Operations 7

operation description vexity slope implicit
constraint

norm(x, p) (
∑︀

𝑥𝑝
𝑖 )

1/𝑝 convex increasing on 𝑥 ≥ 0
decreasing on 𝑥 ≤ 0

p = 1, 2,
Inf

vecnorm(x,
p)

(
∑︀

𝑥𝑝
𝑖𝑗)

1/𝑝 convex increasing on 𝑥 ≥ 0
decreasing on 𝑥 ≤ 0

p = 1, 2,
Inf

quad_form(x,
P)

𝑥𝑇𝑃𝑥 convex in 𝑥
affine in 𝑃

increasing on 𝑥 ≥ 0
decreasing on 𝑥 ≤ 0
increasing in 𝑃

either 𝑥 or 𝑃
must be
constant

quad_over_lin(x,
y)

𝑥𝑇𝑥/𝑦 convex increasing on 𝑥 ≥ 0
decreasing on 𝑥 ≤ 0
decreasing in 𝑦

𝑦 > 0

sum_squares(x)
∑︀

𝑥2
𝑖 convex increasing on 𝑥 ≥ 0

decreasing on 𝑥 ≤ 0
none

sqrt(x)
√
𝑥 convex decreasing 𝑥 > 0

square(x),
x^2

𝑥2 convex increasing on 𝑥 ≥ 0
decreasing on 𝑥 ≤ 0

none

geo_mean(x,
y)

√
𝑥𝑦 concave increasing 𝑥 ≥ 0, 𝑦 ≥ 0

huber(x)
huber(x,
M)

{︃
𝑥2 |𝑥| ≤ 𝑀

2𝑀 |𝑥| −𝑀2 |𝑥| > 𝑀
convex increasing on 𝑥 ≥ 0

decreasing on 𝑥 ≤ 0
𝑀 >= 1

1.4.3 Exponential Cone Representable Functions

An optimization problem using these functions can be solved by any exponential cone solver (SCS).

operation description vexity slope implicit constraint
logsumexp(x) log(

∑︀
𝑖 exp(𝑥𝑖)) convex increasing none

exp(x) exp(𝑥) convex increasing none
log(x) log(𝑥) concave increasing 𝑥 > 0
entropy(x)

∑︀
𝑖𝑗 −𝑥𝑖𝑗 log(𝑥𝑖𝑗) concave not monotonic 𝑥 > 0

logistic_loss(x) log(1 + exp(𝑥𝑖)) convex increasing none

1.4.4 Semidefinite Program Representable Functions

An optimization problem using these functions can be solved by any SDP solver (including SCS and Mosek).

operation description vexity slope implicit constraint
nuclear_norm(x) sum of singular values of 𝑥 convex not monotonic none
operator_norm(x) max of singular values of 𝑥 convex not monotonic none
lambda_max(x) max eigenvalue of 𝑥 convex increasing x is positive semidefinite
lambda_min(x) min eigenvalue of 𝑥 concave increasing x is positive semidefinite

1.4.5 Exponential + SDP representable Functions

An optimization problem using these functions can be solved by any solver that supports exponential constraints and
semidefinite constraints simultaneously (SCS).

operation description vexity slope implicit constraint
logdet(x) log of determinant of 𝑥 concave increasing x is positive semidefinite

8 Chapter 1. In Depth Documentation:

1.4.6 Promotions

When an atom or constraint is applied to a scalar and a higher dimensional variable, the scalars are promoted. For
example, we can do max(x, 0) gives an expression with the shape of x whose elements are the maximum of the
corresponding element of x and 0.

1.5 Examples

• Basic usage

• Control

• Tomography

• Time series

• Section allocation

• Binary knapsack

• Entropy maximization

• Logistic regression

1.6 Solvers

Convex.jl transforms each problem into an equivalent cone program in order to pass the problem to a specialized
solver. Depending on the types of functions used in the problem, the conic constraints may include linear, second-
order, exponential, or semidefinite constraints, as well as any binary or integer constraints placed on the variables.

By default, Convex.jl does not install any solvers. Many users use the solver SCS, which is able to solve problems
with linear, second-order cone constraints (SOCPs), exponential constraints and semidefinite constraints (SDPs). Any
other solver in JuliaOpt may also be used, so long as it supports the conic constraints used to represent the problem.
Most other solvers in the JuliaOpt ecosystem can be used to solve (mixed integer) linear programs (LPs and MILPs).
Mosek and Gurobi can be used to solve SOCPs (even with binary or integer constraints), and Mosek can also solve
SDPs. For up-to-date information about solver capabilities, please see the table here describing which solvers can
solve which kind of problems.

Installing these solvers is very simple. Just follow the instructions in the documentation for that solver.

To use a specific solver, you can use the following syntax

solve!(p, GurobiSolver())
solve!(p, MosekSolver())
solve!(p, GLPKSolverMIP())
solve!(p, GLPKSolverLP())
solve!(p, ECOSSolver())
solve!(p, SCSSolver())

(Of course, the solver must be installed first.) For example, we can use GLPK to solve a MILP

using GLPKMathProgInterface
solve!(p, GLPKSolverMIP())

You can set or see the current default solver by

1.5. Examples 9

get_default_solver()
using Gurobi
set_default_solver(GurobiSolver()) # or set_default_solver(SCSSolver(verbose=0))
# Now Gurobi will be used by default as a solver

Many of the solvers also allow options to be passed in. More details can be found in each solver’s documentation.

For example, if we wish to turn off printing for the SCS solver (ie, run in quiet mode), we can do so by

using SCS
solve!(p, SCSSolver(verbose=false))

If we wish to increase the maximum number of iterations for ECOS or SCS, we can do so by

using ECOS
solve!(p, ECOSSolver(maxit=10000))
using SCS
solve!(p, SCSSolver(max_iters=10000))

1.7 FAQ

• Where can I get help?
• How does Convex.jl differ from JuMP?
• Where can I learn more about Convex Optimization?
• Are there similar packages available in other languages?
• How does Convex.jl work?
• How do I cite this package?

1.7.1 Where can I get help?

For usage questions, please contact us via the JuliaOpt mailing list. If you’re running into bugs or have feature requests,
please use the Github Issue Tracker.

1.7.2 How does Convex.jl differ from JuMP?

Convex.jl and JuMP are both modelling languages for mathematical programming embedded in Julia, and both in-
terface with solvers via the MathProgBase interface, so many of the same solvers are available in both. Convex.jl
converts problems to a standard conic form. This approach requires (and certifies) that the problem is convex and DCP
compliant, and guarantees global optimality of the resulting solution. JuMP allows nonlinear programming through
an interface that learns about functions via their derivatives. This approach is more flexible (for example, you can
optimize non-convex functions), but can’t guarantee global optimality if your function is not convex, or warn you if
you’ve entered a non-convex formulation.

For linear programming, the difference is more stylistic. JuMP’s syntax is scalar-based and similar to AMPL
and GAMS making it easy and fast to create constraints by indexing and summation (like sum{x[i],
i=1:numLocation}). Convex.jl allows (and prioritizes) linear algebraic and functional constructions (like
max(x,y) < A*z); indexing and summation are also supported in Convex.jl, but are somewhat slower than in
JuMP. JuMP also lets you efficiently solve a sequence of problems when new constraints are added or when coeffi-
cients are modified, whereas Convex.jl parses the problem again whenever the solve! method is called.

1.7.3 Where can I learn more about Convex Optimization?

See the freely available book Convex Optimization by Boyd and Vandenberghe for general background on convex
optimization. For help understanding the rules of Disciplined Convex Programming, we recommend this DCP tutorial
website.

1.7.4 Are there similar packages available in other languages?

Indeed! You might use CVXPY in Python, or CVX in Matlab.

1.7.5 How does Convex.jl work?

For a detailed discussion of how Convex.jl works, see our paper.

1.7.6 How do I cite this package?

If you use Convex.jl for published work, we encourage you to cite the software using the following BibTeX citation:

@article{convexjl,
title = {Convex Optimization in {J}ulia},
author ={Udell, Madeleine and Mohan, Karanveer and Zeng, David and Hong, Jenny and Diamond, Steven and Boyd, Stephen},
year = {2014},
journal = {SC14 Workshop on High Performance Technical Computing in Dynamic Languages},
archivePrefix = "arXiv",
eprint = {1410.4821},
primaryClass = "math-oc",

}

